

matrix get — Access system matrices

[Syntax](#)[Description](#)[Remarks and examples](#)[Also see](#)

Syntax

Obtain copy of internal Stata system matrix

```
matrix [define] matname = get(systemname)
```

*Post matrix as internal **Rr** matrix*

```
mat_put_rr matname
```

where *systemname* is

- `_b` coefficients after any estimation command
- `VCE` covariance matrix of estimators after any estimation command
- `Rr` constraint matrix after `test`; see [\[R\] test](#)
- `Cns` constraint matrix after any estimation command

Description

The `get()` matrix function obtains a copy of an internal Stata system matrix. Some system matrices can also be obtained more easily by directly referring to the returned result after a command. In particular, the coefficient vector can be referred to as `e(b)`, the variance–covariance matrix of estimators as `e(V)`, and the constraints matrix as `e(Cns)` after an estimation command.

`mat_put_rr` is a programmer’s command that posts *matname* as the internal **Rr** matrix. *matname* must have one more than the number of columns in the `e(b)` or `e(V)` matrices. The extra column contains the *r* vector, and the earlier columns contain the **R** matrix for the Wald test

$$Rb = r$$

The `matrix ...get(Rr)` command provides a way to obtain the current **Rr** system matrix.

Remarks and examples

[stata.com](#)

`get()` obtains copies of matrices containing coefficients and the covariance matrix of the estimators after estimation commands (such as `regress` and `probit`) and obtains copies of matrices left behind by other Stata commands. The other side of `get()` is `ereturn post`, which allows ado-file estimation commands to post results to Stata’s internal areas; see [\[P\] ereturn](#).

▷ Example 1

After any model-fitting command, the coefficients are available in `_b` and the variance–covariance matrix of the estimators in `VCE`.

```
. use http://www.stata-press.com/data/r13/auto
(1978 Automobile Data)
. regress price weight mpg
(output omitted)
```

Here we can directly use `e(b)` and `e(V)` to obtain the matrices:

```
. matrix list e(b)
e(b)[1,3]
      weight      mpg      _cons
y1  1.7465592 -49.512221  1946.0687
. matrix list e(V)
symmetric e(V)[3,3]
      weight      mpg      _cons
weight  .41133468
      mpg  44.601659  7422.863
_cons -2191.9032 -292759.82  12938766
```

We can also use the `matrix get()` function to obtain these matrices:

```
. matrix b = get(_b)
. matrix V = get(VCE)
. matrix list b
b[1,3]
      weight      mpg      _cons
y1  1.7465592 -49.512221  1946.0687
. matrix list V
symmetric V[3,3]
      weight      mpg      _cons
weight  .41133468
      mpg  44.601659  7422.863
_cons -2191.9032 -292759.82  12938766
```

The columns of `b` and both dimensions of `V` are properly labeled.

◀

▷ Example 2

After `test`, the restriction matrix is available in `Rr`. Having just estimated a regression of `price` on `weight` and `mpg`, we will run a test and then get the restriction matrix:

```
. test weight=1, notest
( 1) weight = 1
. test mpg=40, accum
( 1) weight = 1
( 2) mpg = 40
      F( 2, 71) = 6.29
      Prob > F = 0.0030
. matrix rxtr=get(Rr)
```

```
. matrix list rxtr
rxtr[2,4]
      c1  c2  c3  c4
r1    1   0   0   1
r2    0   1   0  40
```

◀

Also see

- [P] [matrix](#) — Introduction to matrix commands
- [U] [13.5 Accessing coefficients and standard errors](#)
- [U] [14 Matrix expressions](#)